

Application Note AN3201-02: Pitch Shifting **By Frank Thomson**

This application note describes a basic pitch shifting routine that can be implemented in the AL3201.

1. General Explanation

The AL3201 is capable of performing a basic pitch shifting routine by either reading faster (pitch up) or slower (pitch down) through the samples. This is accomplished by running the samples through a circular buffer and using 2 pointers to read from the circular buffer. The 2 pointers are 180 degrees apart in the circular buffer so that as one pointer is crossing from the end to the start of the buffer, the other pointer is mid-way through the buffer.

A sample is calculated by first linearly interpolating between the samples at one pointer, then at the other pointer, then cross-fading between the 2 pointers such that the cross fade coefficient is 0 for the pointer as it crosses from the end to the start of the circular buffer.

The cross fade coefficient is calculated by the digital LFO's along with the sawtooth wave forms required to generate the 2 pointers.

2. Calculations

The digital LFO's are calculated using 24-bit math in the chorus generator block. The 20 MSB's of the results are fed out for use by the address generator and MAC. Of these 20-bits, the 13-MSB's are routed to the address generator and the 7 LSB's are used as the interpolation coefficient for the MAC.

The 13-bits routed to the address generator are 2's comp, giving an effective range of +4095 to -4096. This range is controlled by the amplitude coefficient for the LFO which is 15-bits and is always less than 1.

The frequency is controlled by a 13-bit coefficient. These 13-bits are the LSB's of an 18-bit word used internal to the digital LFO's.

The sawtooth wave frequency is calculated by (assuming a 12.288MHz crystal is being used which results in a 48KHz sample rate) by:

$f = (K * S * C) / (2 * SIN)$ where:

SIN = 8388607 (max positive 24-bit value)

K = 13-bit frequency coefficient / 262143 (262143 is the max an 18-bit value can be internally)

S = 48000 (sample rate)

C = 4194304 (0x400000, a constant used internal in the LFO)

For a frequency coefficient of 1, $K = 1/262143$ and results in $f = 0.045777...Hz$

For a frequency coefficient of 8191 (max 13-bit value), $K = 8191/262143$ and results in $f = 374.9557Hz$

3. Pitch up

Calculating the proper frequency and amplitude coefficient for pitch shifting is a dependent on both length of the circular buffer and the amount of pitch shift desired. We select the buffer to be 8192 samples long (maximum amplitude of the sawtooth waveform) which results in an amplitude coefficient of 31767 (max 15-bit amplitude coefficient).

To determine the frequency coefficient, we first need to determine the amount of shift desired. For the case of pitching up 1 octave, we want to play the 8192 samples from the circular buffer at an effective rate of double the sample rate. An easy way to calculate this is to determine how much faster we want to play the samples than the base sample rate. In this case, we want to play the 8192 samples at a rate that is effectively 48KHz added the base sample rate of 48KHz. Since the samples are written to the circular buffer using addresses generated by a down counter and as the sawtooth is generated by an up counter we need to invert the sawtooth waveform for the pitch up case. This is done by using the COMPA flag in a CHR instruction. Therefore, to calculate the frequency coefficient:

$$8192/48000 = 0.170666... \text{ seconds}$$

$$1/0.170666 = 5.859375 \text{ Hz}$$

From $f=(K*S*C)/(2*SIN)$ above, we can rearrange the equation to solve for K:

$$K = (2*SIN*f)/(S*C)$$

Resulting in $K = 0.00048828...$

As $K = \text{frequency coefficient}/262143$, we rearrange to solve for frequency coefficient:

$$\text{frequency coefficient} = K*262143$$

Resulting in 128.

Therefore, to pitch up one octave, frequency coefficient = 128, amplitude coefficient = 31767 and the buffer length would be 8192.

To shift by up 1/2 octave, using the same 8192 buffer size, we would want to play the sample at a rate that is effectively 19.882251KHz on top of the base rate, therefore:

$$8192/19.88225...KHz = 0.412025... \text{ seconds}$$

$$1/0.412025... = 2.42703... \text{ Hz}$$

Solving for K:

$$K = 0.000202252.....$$

Solving for the frequency coefficient:

$$\text{Frequency coefficient} = 53$$

```

;
; File:          AN320102U.ASM
; Description:   Example pitch up program
; Authors:      Frank Thomson
; Copyright 2001 Alesis Semiconductor
;
;
; 1 octave up
LFO0   SAW   AMP=32767   FREQ=128   XFAD=1/16
;
;NOTE: Here we define the waveform:
;     SIN, TRI (triangle), SAW (sawtooth for pitch shift)
;
;
MEM     pitchmem      8200           ; Circular buffer memory
MEM     temp          1             ; temp register
MEM     temp2         1             ; temp register
;NOTE: memory locations are referenced by:
;     name Start of memory block
;     name' End of memory block
;     name" Middle of memory block
;
RZP     ADCL          K=.999        ;0   Read left channel into Accumulator
;
WZP     pitchmem      K=.999        ;1   Write to the start of the memory
block
; Next read from the middle of the delay memory block using the chorus instruction to
add the
; sawtooth waveform to the address. The sample returned is multiplied by the 1's comp of
the
; coefficient from the chorus generator block. Use the SIN output for the first
sawtooth. As this
; is a pitch up, 1's comp the address.
;
CHR0 RZP pitchmem"   +SIN  COMPA  COMPK  LATCH  ;2
;
; Read the sample before the above one and multiply it by the coefficient.
;
CHR0 RAP pitchmem"+1 +SIN  COMPA          ;3   Read middle -1 from memory block
;
; Save the result to a temp register.
;
WAP     temp          K=0           ;4   Save result to temp location
;
; Now do it again using the COS output to get the other sawtooth
; Again inverting the address portion of the sawtooth
;
CHR0 RZP pitchmem"   +COS  COMPA  COMPK  LATCH  ;5
CHR0 RAP pitchmem"+1 +COS  COMPA          ;6
;
; The result from the 2 instructions is in the accumulator, do a write to a temp
location to bring it
; back through the multiplier and use the 1's comp of the cross fade coefficient to
multiply it
; by.
;
CHR0 WZP temp2 +COS MASKA COMPK          ;7
;
; Get the first result, multiply it by the cross fade coefficient and add it.

```



```

;
CHRO RAP temp MASKA ;8
;
WAP OUTL K=0 ;9 Write it to the DAC
;
;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
;
RZP ADCL K=.999 ;10 Read left channel into Accumulator
WAP OUTR K=0 ;11 Write to right output
;
;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
;Add a bunch of reads for refresh
RZP 0x00 K=0
RZP 0x40 K=0
RZP 0x80 K=0
RZP 0xc0 K=0
RZP 0x100 K=0
RZP 0x140 K=0
RZP 0x180 K=0
RZP 0x1c0 K=0
RZP 0x200 K=0
RZP 0x240 K=0
RZP 0x280 K=0
RZP 0x2c0 K=0
RZP 0x300 K=0
RZP 0x340 K=0
RZP 0x380 K=0
RZP 0x3c0 K=0

```

4. Pitch down

Pitching down is similar to pitching up, except rather than adding to the base sample rate, we are subtracting from it. In this case, we do not invert the sawtooth waveform. As an example, if we wanted to pitch down by 1 octave we effectively want to subtract 24KHz from the sample rate. Using the same 8192 sample buffer, we calculate as follows:

$$8192/24000 = 0.341333...$$

$$1/0.341333... = 2.9296875 \text{ Hz}$$

Solving for K:

$$K = 0.00024366...$$

Solving for the frequency coefficient:

$$\text{Frequency coefficient} = 64$$

As another example, shifting down 1/2 octave, we effectively want to subtract 14.05887...KHz from the base sample rate of 48KHz,

$$8192/14.05887\text{KHz} = 0.5826924...\text{Hz}$$

$$1/0.5826924... = 1.716171...\text{Hz}$$



Solving for K:

K = 0.00014301425

Solving for the frequency coefficient:

Frequency coefficient = 37

```

;
; File:          AN320102D.ASM
; Description:   Example pitch down program
; Authors:      Frank Thomson
; Copyright 2001 Alesis Semiconductor
;
;
; ~~~~~
; 1 octave down
LFO0  SAW  AMP=32767  FREQ=64  XFAD=1/16
;NOTE: Here we define the waveform:
;     SIN, TRI (triangle), SAW (sawtooth for pitch shift)
;
;
MEM   pitchmem      8200          ; Circular buffer memory
MEM   temp           1           ; Temp register
MEM   temp2          1           ; Temp register
;NOTE: memory locations are referenced by:
;     name Start of memory block
;     name' End of memory block
;     name" Middle of memory block
;
; ~~~~~
RZP   ADCL           K=.999      ;0  Read left channel into Accumulator
;
WZP   pitchmem      K=.999      ;1  Write to the start of the block
; Next read from the middle of the delay memory block using the chorus instruction to
add the
; sawtooth waveform to the address. The sample returned is multiplied by the 1's comp of
the
; coefficient from the chorus generator block. Use the SIN output for the first
sawtooth.
;
CHR0  RZP pitchmem"  +SIN  COMPK  LATCH  ;2  Read middle of memory block
;
; Read the sample before the above one and multiply it by the coefficient.
;
CHR0  RAP pitchmem"+1  +SIN          ;3  Read middle -1 from memory block
;
; Save the result to a temp register.
;
WAP   temp          K=0         ;4  Write to DAC
;
; Now do it again using the COS output to get the other sawtooth
;
CHR0  RZP pitchmem"  +COS  COMPK  LATCH  ;5
CHR0  RAP pitchmem"+1  +COS          ;6
;
; The result from the 2 instructions is in the accumulator, do a write to a temp
location to bring it
; back throught the multiplier and use the 1's comp of the cross fade coefficient to
multiply it
; by.

```

AN3201-02-0702 Alesis Semiconductor
12555 Jefferson Blvd., Suite 285
Los Angeles, CA 90066
Phone (310) 301-0780 Fax (310) 306-1551 www.alesis-semi.com



```

;
CHR0 WZP temp2 +COS COMPK MASKA          ;7
;
; Get the first result, multiply it by the cross fade coefficient and add it.
;
CHR0 RAP temp MASKA                      ;8
;
WAP OUTL K=0                             ;9  Write it to the DAC
;
;
;
RZP  ADCL          K=.999          ;10  Read left channel into Accumulator
WAP  OUTR          K=0              ;11  Write to DAC
;
;
;Add a bunch of reads for refresh
RZP  0x00          K=0
RZP  0x40          K=0
RZP  0x80          K=0
RZP  0xc0          K=0
RZP  0x100         K=0
RZP  0x140         K=0
RZP  0x180         K=0
RZP  0x1c0         K=0
RZP  0x200         K=0
RZP  0x240         K=0
RZP  0x280         K=0
RZP  0x2c0         K=0
RZP  0x300         K=0
RZP  0x340         K=0
RZP  0x380         K=0
RZP  0x3c0         K=0

```